[c1]

Towards a miniKanren with fair search strategies

KUANG-CHEN LU, Indiana University
WEIXI MA, Indiana University
DANIEL P. FRIEDMAN, Indiana University

We describe fairness levels in disjunction and conjunction implementations. Specifically, a disjunction implementation can be fair, almost-fair, or unfair. And a conjunction implementation can be fair or unfair. We compare the fairness level of four search strategies: the standard miniKanren interleaving depth-first search (DFS_i), the balanced interleaving depth-first search (DFS_{bi}), the fair depth-first search (DFS_f), and the standard breadth-first search (BFS_{ser}). DFS_{bi} and DFS_f are new. And we present a new, more efficient and simpler implementation of BFS_{ser}. Using quantitative evaluation, we argue that DFS_{bi} and DFS_f are competitive alternatives to DFS_i, and that BFS_{opt} is more efficient than the standard BFS_{ser}.

ACM Reference Format:
Kuang-Chen Lu, Weixi Ma, and Daniel P. Friedman. 2019. Towards a miniKanren with fair search strategies. 1, 1 (May 2019), 14 pages. https://doi.org/10.1145/nnnnnnn.nnnnnnn

1 INTRODUCTION

miniKanren is a family of relational programming languages. Friedman et al. [2] introduce miniKanren and its implementation in *The Reasoned Schemer, 2nd Ed* (TRS2). miniKanren programs, have been proven to be useful in solving many problems as in miniKanren.org.Byrd et al. [1].

[c2]

[c3]

[c4]

[c5] [c6] A subtlety arises when a `cond`^{e} contains many clauses: not every clause has an equal chance to contribute to the result. As an example, consider the following relation `repeato` and its invocation.

[c1] *LKC*: submission deadline: Mon 27 May 2019
[c2] *DPF*: I think that the abbreviations for the search strategies should not come up until the Introduction.
[c3] *DPF*: We should reference the earliest paper (See Will Byrd vita)
[c4] *DPF*: When we talk about complete and incomplete, I think we should make it explicit that we are only talking about (run n (x y z) g ...) and not (run* (x y z) g ...), but an alternative is that a stream is *productive* or *unproductive*.
[c5] *LKC*: Shall we delete the sentence about relational interpreters? It is misleading: it is emphasizing relational interpreters, but we say nothing further about them later.
[c6] *DPF*: I have deleted the sentence.

Authors' addresses: Kuang-Chen LuIndiana University; Weixi MaIndiana University; Daniel P. FriedmanIndiana University.

```
(defrel (repeat^o x out)
  (cond^e
    ((≡ `(,x) out))
    ((fresh (res)
       (≡ `(,x . ,res) out)
       (repeat^o x res)))))
> (run 4 q
    (repeat^o '* q))
'((*) (* *) (* * *) (* * * *))
```

Next, consider the following disjunction of invoking `repeato` with four different letters.

```
> (run 12 q
    (cond^e
      ((repeat^o 'a q))
      ((repeat^o 'b q))
      ((repeat^o 'c q))
      ((repeat^o 'd q))))
```

cond^e intuitively relates its clauses with logical `or`. And thus an unsuspicious beginner would expect each letter to contribute equally to the result, as follows.

```
'((a) (b) (c) (d)
  (a a) (b b) (c c) (d d)
  (a a a) (b b b) (c c c) (d d d))
```

The cond^e in TR2, however, generates a less expected result.

```
'((a) (a a) (b) (a a a)
  (a a a a) (b b)
  (a a a a a) (c)
  (a a a a a a) (b b b)
  (a a a a a a a) (d))
```

The miniKanren in TRS2 implements interleaving DFS (DFS$_i$), the cause of this unexpected result. With this search strategy, each clause takes half of its received computational resources and passes the other half to its following clauses, except for the last clause that takes all resources it receives. In the example above, the `a` clause takes half of all resourses. And the `b` clause takes a quarter. Thus `c` and `d` barely contribute to the result.

DFS$_i$ is sometimes powerful for an expert. By carefully organizing the order of cond^e clauses, a miniKanren program can explore more "interesting" clauses than those uninteresting ones, and thus use computational resources efficiently. A little miniKanrener, however, may beg to differ–understanding implementation details and fiddling with clause order is not the first priority of a beginner.

There is another reason that miniKanren could use more search strategies than just DFS$_i$. In many applications, there does not exist one order that serves all purposes. For example, a relational dependent type checker contains clauses for constructors that build data and clauses for eliminators that use data. When the type checker is used to generate simple and shallow programs, the clauses for constructors should be put at the top of the cond^e clauses. When performing proof searches for complicated programs, the clauses for eliminators should be put at the top of cond^e clauses. With DFS$_i$, these two uses cannot be efficient at the same time. In fact, to make one use efficient, the other one must be more sluggish.

The specification that every clause in the same cond^e is given equal "search priority" is called fair `disj`. And search strategies with almost-fair `disj` give every clause similar priority. Fair `conj`, a related concept, is more subtle. The discussion of fair `conj` is covered in the next section.

To summarize our contribution, we

- propose and implement balanced interleaving depth-first search (DFS_{bi}), a new search strategy with almost-fair `disj`.
- propose and implement fair depth-first search (DFS_f), a new search strategy with fair `disj`.
- implement in a new way breath-first search (BFS_{opt}), a search strategy with fair `disj` and fair `conj`. And we prove formally that our BFS_{opt} implementation is semantically equivalent to the one by Seres et al. [5], however, Our code runs faster in all benchmarks and we believe it is simpler.

2 SEARCH STRATEGIES AND FAIRNESS

In this section, we define fair `disj`, almost-fair `disj` and fair `conj`. Before going further into fairness, we give a short review of the terms: *state*, search *space*, *goal*. A *state* is a collection of constraints. (Here, we restrict constraints to unification constraints.) Every answer corresponds to a state. A *space* is a collection of states. And a *goal* is a function from a state to a space.

Now we elaborate fairness by running more queries about `repeato`.

2.1 fair `disj`

Given the following program, it is natural to expect lists of each letter to constitute 1/4 in the query result. DFS_i, the current search strategy, however, results in many more lists of as than lists of other letters. And some letters (e.g. c and d) are rarely seen. The situation would be exacerbated if cond^e would have contained more clauses.

```
;; \DFSi\ (unfair disj)
> (run 12 q
    (cond^e
      ((repeat^o 'a q))
      ((repeat^o 'b q))
      ((repeat^o 'c q))
      ((repeat^o 'd q))))
'((a) (a a) (b) (a a a)
  (a a a a) (b b)
  (a a a a a) (c)
  (a a a a a a) (b b b)
  (a a a a a a a) (d))
```

Under the hood, the cond^e here is allocating computational resources to four trivially different search spaces. The unfair `disj` in DFS_i allocates many more resources to the first search space. On the contrary, fair `disj` would allocate resources evenly to each search space.

```
;; \DFSf{} (fair disj)
> (run 12 q
    (cond^e
      ((repeat^o 'a q))
      ((repeat^o 'b q))
      ((repeat^o 'c q))
      ((repeat^o 'd q))))
'((a) (b) (c) (d)
  (a a) (b b) (c c) (d d)
  (a a a) (b b b) (c c c) (d d d))
```

```
;; BFSser (fair disj)
> (run 12 q
    (cond^e
      ((repeat^o 'a q))
      ((repeat^o 'b q))
      ((repeat^o 'c q))
      ((repeat^o 'd q))))
'((a) (b) (c) (d)
  (a a) (b b) (c c) (d d)
  (a a a) (b b b) (c c c) (d d d))
```

Running the same program again with almost-fair `disj` (e.g. DFS_{bi}) gives the same result. Almost-fair, however, is not completely fair, as shown by the following example.

```
;; DFSbi (almost-fair disj)
> (run 16 q
    (cond^e
      ((repeat^o 'a q))
      ((repeat^o 'b q))
      ((repeat^o 'c q))
      ((repeat^o 'd q))
      ((repeat^o 'e q))))
'((b) (c) (d) (a)
  (b b) (c c) (d d) (e)
  (b b b) (c c c) (d d d) (a a)
  (b b b b) (c c c c) (d d d d) (e e))
```

DFS_{bi} is fair only when the number of goals is a power of 2, otherwise, some goals would be allocated twice as many resources than the others. In the above example, where the cond^e has five clauses, the clauses of `b`, `c`, and `d` are allocated more resources.

We end this subsection with precise definitions of all levels of `disj` fairness. Our definition of *fair* `disj` is slightly more general than the one in Seres et al. [5]. Their definition is only for binary disjunction. We generalize it to a multi-arity one.

Definition 2.1 (fair `disj`). *A* `disj` *is fair if and only if it allocates computational resources evenly to search spaces produced by goals in the same disjunction (i.e., clauses in the same* cond^e*).*

Definition 2.2 (almost-fair `disj`). *A* `disj` *is almost-fair if and only if it allocates computational resources evenly to search spaces produced by goals in the same disjunction that the maximal ratio of resources is bounded by a constant.*

Definition 2.3 (unfair `disj`). *A* `disj` *is unfair if and only if it is not almost-fair.*

2.2 fair conj

c1

[c1] *LKC*: Checked grammer of text before this line

Given the following program, it is natural to expect lists of each letter to constitute 1/4 in the answer. Search strategies with unfair `conj` (e.g. DFS_i, DFS_{bi}, DFS_f), however, results in many more lists of `a`s than lists of other letters. And some letters are rarely seen. The situation would be exacerbated if `cond`^{e} were to contain more clauses. Although DFS_i's `disj` is unfair in general, it is fair when there is no call to a relational definition in `cond`^{e} clauses (including this case).

```
;; DFSi (unfair conj)
> (run 12 q
    (fresh (x)
      (cond^e
        ((≡ 'a x))
        ((≡ 'b x))
        ((≡ 'c x))
        ((≡ 'd x)))
      (repeat^o x q)))
'((a) (a a) (b) (a a a)
  (a a a a) (b b)
  (a a a a a) (c)
  (a a a a a a) (b b b)
  (a a a a a a a) (d))
```

```
;; DFSf (unfair conj)
> (run 12 q
    (fresh (x)
      (cond^e
        ((≡ 'a x))
        ((≡ 'b x))
        ((≡ 'c x))
        ((≡ 'd x)))
      (repeat^o x q)))
'((a) (a a) (b) (a a a)
  (a a a a) (b b)
  (a a a a a) (c)
  (a a a a a a) (b b b)
  (a a a a a a a) (d))
```

```
;; DFSbi (unfair conj)
> (run 12 q
    (fresh (x)
      (cond^e
        ((≡ 'a x))
        ((≡ 'b x))
        ((≡ 'c x))
        ((≡ 'd x)))
      (repeat^o x q)))
'((a) (a a) (c) (a a a)
  (a a a a) (c c)
  (a a a a a) (b)
  (a a a a a a) (c c c)
  (a a a a a a a) (d))
```

Under the hood, the `cond`^{e} and the call to `repeato` are connected by `conj`. The `cond`^{e} goal outputs a search space including four trivially different states. Then the next conjunctive goal, `(repeato x q)`, is applied to each of these states, producing four trivially different search spaces. In the examples above, the `conj`s are allocating more computational resources to the search space of `a`. On the contrary, fair `conj` would allocate resources evenly to each search space. For example,

```
;; \BFSser{} (fair conj)
> (run 12 q
    (fresh (x)
      (cond^e
        ((≡ 'a x))
        ((≡ 'b x))
        ((≡ 'c x))
        ((≡ 'd x)))
      (repeat^o x q)))
'((a) (b) (c) (d)
  (a a) (b b) (c c) (d d)
  (a a a) (b b b) (c c c) (d d d))
```

A more interesting situation is when the first conjunct produces an unbounded number of states. Consider the following example, a naive specification of fair `conj` might require search strategies to produce all sorts of singleton lists, but no longer ones, which makes the strategies incomplete. A search strategy is *complete* if and only if it can determine all the answers within finite time, otherwise, it is *incomplete*.

```
;; naively fair conj
> (run 6 q
    (fresh (xs)
      (conde
        ((repeato 'a xs))
        ((repeato 'b xs)))
      (repeato xs q)))
'(((a)) ((b))
  ((a a)) ((b b))
  ((a a a)) ((b b b)))
```

Our solution requires a search strategy with *fair* `conj` to organize states in bags in search spaces, where each bag contains finite states, and to allocate resources evenly among search spaces derived from states in the same bag. It is up to a search strategy designer to decide by what criteria to put states in the same bag, and how to allocate resources among search spaces related to different bags.

BFS_{ser} puts states of the same cost in the same bag, and allocates resources carefully among search spaces related to different bags such that answers are produced in increasing order of cost. The *cost* of an answer is its depth in the search tree (i.e., the number of calls to relations required to find them) [5]. In the following example, every answer is a list of list of symbols, where inner lists are the same. Here the cost of each answer is equal to the length of its inner lists plus the length of its outer list.

```
;; BFSser (fair conj)
> (run 12 q
    (fresh (xs)
      (conde
        ((repeato 'a xs))
        ((repeato 'b xs)))
      (repeato xs q)))
'(((a)) ((b))
  ((a) (a)) ((b) (b))
  ((a a)) ((b b))
  ((a) (a) (a)) ((b) (b) (b))
  ((a a) (a a)) ((b b) (b b))
  ((a a a)) ((b b b)))
```


We end this subsection with precise definitions of all levels of `conj` fairness.

Definition 2.4 (fair `conj`). *A* `conj` *is fair if and only if it allocates computational resources evenly to search spaces produced from states in the same bag. A bag is a finite collection of states. And search strategies with fair* `conj` *should represent search spaces with possibly unbounded collections of bags.*

Definition 2.5 (unfair `conj`). *A* `conj` *is unfair if and only if it is not fair.*

3 INTERLEAVING DEPTH-FIRST SEARCH

In this section, we review the implementation of interleaving depth-first search (DFS_{i}). We focus on parts that are relevant to this paper. TRS2 [2] provides a comprehensive description of the whole miniKanren implementation.

```
(define-syntax cond^e
  (syntax-rules ()
    ((cond^e (g ...) ...)
     (disj (conj g ...) ...))))
```

Fig. 1. implementation of cond^e

```
#| Goal × Goal → Goal |#
(define (disj2 g1 g2)
  (lambda (s)
    (append^∞ (g1 s) (g2 s))))

#| Space × Space → Space |#
(define (append^∞ s^∞ t^∞)
  (cond
    ((null? s^∞) t^∞)
    ((pair? s^∞)
     (cons (car s^∞)
       (append^∞ (cdr s^∞) t^∞)))
    (else (lambda ()
            (append^∞ t^∞ (s^∞))))))

(define-syntax disj
  (syntax-rules ()
    ((disj) (fail))
    ((disj g0 g ...) (disj+ g0 g ...))))

(define-syntax disj+
  (syntax-rules ()
    ((disj+ g) g)
    ((disj+ g0 g1 g ...) (disj2 g0 (disj+ g1 g ...)))))
```

Fig. 2. implementation of DFS$_i$ (Part I)

The definition of cond^e (Fig. 1) is shared by all search strategies. It relates clauses disjunctively, and goals in the same clause conjunctively.
c1

Fig. 2 and Fig. 3 show parts that are later compared with other search strategies. We follow some conventions to name variables: variables bound to states are named s; variables bound to goals are named g (possibly with subscript); variables bound to search spaces have names ending with $^\infty$. Fig. 2 shows the implementation of

[c1] *DPF*: If we use the name subscript, the subscripts should use an underscore and we should do it for gunderscorel and gundescorer. If we can use an l that looks less like a 1, that would even be better. Look at the l in TLT.

```
#| Goal × Goal → Goal |#
(define (conj2 g1 g2)
  (lambda (s)
    (append-map∞ g2 (g1 s))))

#| Goal × Space → Space |#
(define (append-map∞ g s∞)
  (cond
    ((null? s∞) '())
    ((pair? s∞)
     (append∞ (g (car s∞))
       (append-map∞ g (cdr s∞))))
    (else (lambda ()
            (append-map∞ g (s∞))))))

(define-syntax conj
  (syntax-rules ()
    ((conj) (fail))
    ((conj g0 g ...) (conj+ g0 g ...))))

(define-syntax conj+
  (syntax-rules ()
    ((conj+ g) g)
    ((conj+ g0 g1 g ...) (conj2 g0 (conj+ g1 g ...)))))
```

Fig. 3. implementation of DFS_i (Part II)

`disj`. The first function, `disj2`, implements binary disjunction. It applies the two disjunctive goals to the input state `s` and composes the two resulting search spaces with append$^\infty$. The following syntax definitions say `disj` is right-associative. Fig. 3 shows the implementation of `conj`. The first function, `conj2`, implements binary conjunction. It applies the *first* goal to the input state, then applies the second goal to states in the resulting search space. The latter process is done with a the helper function append-map$^\infty$ that applies its input goal to states in its input search spaces and composes the resulting search spaces. It reuses append$^\infty$ for search space composition. The following syntax definitions say `conj` is also right-associative.

4 BALANCED INTERLEAVING DEPTH-FIRST SEARCH

Balanced interleaving DFS (DFS_{bi}) has an almost-fair `disj` and unfair `conj`. The implementation of DFS_{bi} differs from DFS_i's in the `disj` macro. We list the new `disj` with its helper in Fig. 4. When there are one or more disjunctive goals, `disj` builds a balanced binary tree whose leaves are the goals and whose nodes are `disj2`s, hence the name of this search strategy. The new helper, `disj+`, takes two additional 'arguments'. They accumulate goals to be put in the left and right subtrees. The first clause handles the case where there is only one goal. In this case, the tree is the goal itself. When there are more goals, we partition the list of goals into two sublists of roughly equal lengths and recur on the two sublists. Goals are moved to the accumulators in the last clause. As

```
(define-syntax disj
  (syntax-rules ()
    ((disj) fail)
    ((disj g ...) (disj+ (g ...) () ()))))

(define-syntax disj+
  (syntax-rules ()
    ((disj+ () () g) g)
    ((disj+ (g_l ...) (g_r ...))
     (disj2 (disj+ () () g_l ...)
            (disj+ () () g_r ...)))
    ((disj+ (g_l ...) (g_r ...) g_0)
     (disj2 (disj+ () () g_l ... g_0)
            (disj+ () () g_r ...)))
    ((disj+ (g_l ...) (g_r ...) g_0 g_1 g ...))
     (disj+ (g_l ... g_0) (g_r ... g_1) g ...)))
```

Fig. 4. implementation of DFS_{bi}

we are moving two goals each time, there are two base cases: (1) no goal remains; (2) one goal remains. These two new base cases are handled in the second clause and the third clause, respectively. In contrast, the `disj` in DFS_{i} constructs the binary tree in a particularly unbalanced form.

5 FAIR DEPTH-FIRST SEARCH

c1

Fair DFS (DFS_{f}) has fair `disj` and unfair `conj`. The implementation of DFS_{f} differs from DFS_{i}'s in `disj2` (Fig. 5). `disj2` is changed to call a new and fair version of append$^{\infty}$. append$^{\infty}_{fair}$ immediately calls its helper, `loop`, with the first argument, `s?`, set to `#t`, which indicates that s^{∞} and t^{∞} haven't been swapped. The swapping happens at the third `cond` clause in the helper, where `s?` is updated accordingly. The first two `cond` clauses essentially copy the `cars` and stop recursion when one of the input spaces is obviously finite. The third clause, as we mentioned above, is just for swapping. When the fourth and last clause runs, we know that both s^{∞} and t^{∞} are ended with a thunk and they have been swapped. In this case, a new thunk is constructed. The new thunk swapped two search spaces back before recursion. This is unnecessary for fairness. We do it such that answers are produced in a more natural order.

c2 c3

6 BREADTH-FIRST SEARCH

BFS_{ser} has both fair `disj` and fair `conj`. Our implementation is based on DFS_{f} (not DFS_{i}). To implement BFS_{opt} based on DFS_{f}, we need append-map$^{\infty}_{fair}$ in addition to append$^{\infty}_{fair}$. The only difference between append-map$^{\infty}$ and append-map$^{\infty}_{fair}$ is that the latter calls append$^{\infty}_{fair}$ instead of append$^{\infty}$.

[c1] *DPF*: Why does the pair? case precede the null? case? Is it simply more efficient? It shouldn't be?
[c2] *DPF*: When or where do you swap them back?
[c3] *LKC*: fixed

```
#| Goal × Goal → Goal |#
(define (disj2 g₁ g₂)
  (lambda (s)
    (append^∞_fair (g₁ s) (g₂ s))))

#| Space × Space → Space |#
(define (append^∞_fair s^∞ t^∞)
  (let loop ((s? #t) (s^∞ s^∞) (t^∞ t^∞))
    (cond
      ((null? s^∞) t^∞)
      ((pair? s^∞)
       (cons (car s^∞)
         (loop s? (cdr s^∞) t^∞)))
      (s? (loop #f t^∞ s^∞))
      (else (lambda ()
              (loop #t (t^∞) (s^∞)))))))
```

Fig. 5. implementation of DFS_f

The implementation can be improved in two ways. First, as mentioned in section 2.2, BFS_serputs answers in bags and answers of the same cost are in the same bag. In this implementation, however, it is unclear where this information is recorded. Second, $\text{append}^\infty_{fair}$ is extravagant in memory usage. It makes $O(n + m)$ new cons cells every time, where n and m are the "length"s of input search spaces. We address these issues in the first subsection.

Both BFS_ser Seres et al. [5] and BFSopt produce answers in increasing order of cost. So it is interesting to see if they are equivalent. We prove so in Coq. The details are in the second subsection.

6.1 optimized BFS

c1.

As mentioned in section 2.2, BFS puts answers in bags and answers of the same cost are in the same bag. The cost information is recorded subtly – the cars of a search space have cost 0 (i.e. they are all in the same bag), and the costs of answers in thunk are computed recursively then increased by one. It is even more subtle that $\text{append}^\infty_{fair}$ and the $\text{append-map}^\infty_{fair}$ respects the cost information. We make these facts more obvious by changing the type of search space, modifying related function definitions, and introducing a few more functions.

The new type is a pair whose car is a list of answers (the bag), and whose cdr is either a #f or a thunk returning a search space. A falsy cdr means the search space is obviously finite.

Functions related to the pure subset are listed in Fig. 6 (the others in Fig. 7). They are compared with Seres et al.'s implementation in our proof. The first three functions in Fig. 6 are search space constructors. none makes an empty search space; unit makes a space from one answer; and step makes a space from a thunk. The remaining functions as before.

Luckily, the change in $\text{append}^\infty_{fair}$ also fixes the miserable space extravagance—the use of append helps us to reuse the first bag of t^∞.

c1 *DPF*: Never refer to a literal section like 2.2. Instead stick a label where it starts right near the section number and stick another label where the subsection starts. Then ref the two labels and place a period between them.

```
#| → Space |#
(define (none)   `(()   . #f))

#| State → Space |#
(define (unit s) `((,s) . #f))

#| (↪ Space) → Space |#
(define (step f) `(()   . ,f))

#| Space × Space → Space |#
(define (append^∞_fair s^∞ t^∞)
  (cons (append (car s^∞) (car t^∞))
    (let ((t1 (cdr s^∞)) (t2 (cdr t^∞)))
      (cond
        ((not t1) t2)
        ((not t2) t1)
        (else (lambda ()
                (append^∞_fair (t1) (t2))))))))

#| Goal × Space → Space |#
(define (append-map^∞_fair g s^∞)
  (foldr
    (lambda (s t^∞)
      (append^∞_fair (g s) t^∞))
    (let ((f (cdr s^∞)))
      (step (and f (lambda () (append-map^∞_fair g (f))))))
    (car s^∞)))

#| Maybe Nat × Space → [State] |#
(define (take^∞ n s^∞)
  (let loop ((n n) (vs (car s^∞)))
    (cond
      ((and n (zero? n)) '())
      ((pair? vs)
       (cons (car vs)
         (loop (and n (sub1 n)) (cdr vs))))
      (else
       (let ((f (cdr s^∞)))
         (if f (take^∞ n (f)) '()))))))
```

Fig. 6. new and changed functions in optimized BFS that implements pure features

```
#| Space × (State × Space → Space) × (↪ Space) → Space |#
(define (elim s∞ ks kf)
  (let ((ss (car s∞)) (f (cdr s∞)))
    (cond
      ((and (null? ss) f)
       (step (lambda () (elim (f) ks kf))))
      ((null? ss) (kf))
      (else (ks (car ss) (cons (cdr ss) f))))))

#| Goal × Goal × Goal → Goal |#
(define (ifte g1 g2 g3)
  (lambda (s)
    (elim (g1 s)
      (lambda (s0 s∞)
        (append-map∞fair g2
          (append∞fair (unit s0) s∞)))
      (lambda () (g3 s)))))

#| Goal → Goal |#
(define (once g)
  (lambda (s)
    (elim (g s)
      (lambda (s0 s∞) (unit s0))
      (lambda () (none)))))
```

Fig. 7. new and changed functions in optimized BFS that implements impure features

Kiselyov et al. [3] has shown that a *MonadPlus* hides in implementations of logic programming system. The BFS_{opt} implementation is not an exception: $\texttt{append-map}^{\infty}_{fair}$ is like `bind`, but takes arguments in reversed order; `none`, `unit`, and $\texttt{append}^{\infty}_{fair}$ correspond to `mzero`, `unit`, and `mplus` respectively.

Functions implementing impure features are in Fig. 7. The first function, `elim`, takes a space $\texttt{s}^{\infty}$ and two continuations `ks` and `kf`. When $\texttt{s}^{\infty}$ contains some answers, `ks` is called with the first answer and the rest space. Otherwise, `kf` is called with no argument. Here 's' and 'f' means 'succeed' and 'fail' respectively. This function is like an eliminator of search space, hence the name. The remaining functions do the same thing as before.

6.2 comparison with the BFSser and BFSopt

In this section, we compare the pure subset of BFS_{opt} with the BFS found in Seres et al. [5]. We focus on the pure subset because their system is pure. Their system represents search spaces with streams of lists of answers, where each list is a bag.

To compare efficiency, we translate their Haskell code into Racket (See supplements for the translated code). The translation is direct due to the similarity in both logic programming systems and search space representations.

benchmark	size	DFS_i	DFS_{bi}	DFS_f	BFS_{opt}	BFS_{ser}
`very-recursive`^{o}	100000	579	793	2131	1438	3617
	200000	1283	1610	3602	2803	4212
	300000	2160	2836	-	6137	-
`append`^{o}	100	31	41	42	31	68
	200	224	222	221	226	218
	300	617	634	593	631	622
`revers`^{o}	10	5	3	3	38	85
	20	107	98	51	4862	5844
	30	446	442	485	123288	132159
quine-1	1	71	44	69	-	-
	2	127	142	95	-	-
	3	114	114	93	-	-
quine-2	1	147	112	56	-	-
	2	161	123	101	-	-
	3	289	189	104	-	-
'(I love you)-1	99	56	15	22	74	165
	198	53	72	55	47	74
	297	72	90	44	181	365
'(I love you)-2	99	242	61	16	66	99
	198	445	110	60	42	64
	297	476	146	49	186	322

Table 1. The results of a quantitative evaluation: running times of benchmarks in milliseconds

The translated code is longer and slower than BFSopt implementation. Details about difference in efficiency are in section 7.

We prove in Coq that BFS_{ser} and BFS_{opt} are semantically equivalent, i.e. `(run n ? g)` produces the same result (See supplements for the formal proof).

7 QUANTITATIVE EVALUATION

Here we compare the efficiency of search strategies. A concise description is in Table 1. A hyphen means "running out of memory." The first two benchmarks are taken from TRS2. `revers`^{o} is from Rozplokhas and Boulytchev [4]. The next two benchmarks about quine are modified from a similar test case in Byrd et al. [1]. The modifications are made to circumvent the need for symbolic constraints (e.g. $\neq$, `absent`^{o}). Our version generates de Bruijnized expressions and prevent closures getting into list. The two benchmarks differ in the `cond`^{e} clause order of their relational interpreters. The last two benchmarks are about synthesizing expressions that evaluate to `'(I love you)`. This benchmark is from Byrd et al. [1]. Again, the sibling benchmarks differ in the `cond`^{e} clause order of their relational interpreters. The first one has elimination rules (i.e. application, `car`, and `cdr`) at the end, while the other has them at the beginning. We conjecture that DFS_i would perform badly in the second case because elimination rules complicate the problem when synthesizing (i.e., our evaluation supports our conjecture.)

In general, only DFS_i and DFS_{bi} constantly perform well. DFS_f is just as efficient in all benchmarks but `very-recursive`^{o}. Both BFS have obvious overhead in many cases. Among the three variants of DFS (they all have unfair `conj`), DFS_f is most resistant to clause permutation, followed by DFS_{bi} then DFS_i. Among the two implementation of BFS, ours constantly performs as well or better. Interestingly, every strategy with fair `disj`

suffers in `very-recursive`^{o} and DFS_f performs well elsewhere. Therefore, this benchmark might be a special case. Fair `conj` imposes considerable overhead constantly except in `append`^{o}. The reason might be that strategies with fair `conj` tend to keep more intermediate answers in the memory.

8 RELATED WORKS

Yang Yang [6] points out a disjunct complex would be 'fair' if it were a full and balanced tree .

Seres et al Seres et al. [5] also describe a breadth-first search strategy. We prove BFS_{ser} is semantically equivalent to BFS_{opt}. But our codeis shorter and performs better in comparison with a straightforward translation of their Haskell code.

9 CONCLUSION

We analyze the definitions of fair `disj` and fair `conj`, then propose a new definition of fair `conj`. Our definition is orthogonal with completeness.

We devise two new search strategies (i.e. balanced interleaving DFS (DFS_{bi}) and fair DFS (DFS_f)) and devise a new implementation of BFS. These strategies have different features in fairness: $_{bi}$ has an almost-fair `disj` and unfair `conj`. DFS_f has fair `disj` and unfair `conj`. BFS_{opt} has both fair `disj` and fair `conj`.

Our quantitative evaluation shows that DFS_{bi} and DFS_f are competitive alternatives to DFS_i, the current search strategy, and that $BFS_{opt}$is less practical than other search strategies. [c1] [c2]

We prove that BFS_{opt} is equivalent to BFS_{ser}in Seres et al. [5]. Our optimized code is shorter and runs faster than a direct translation of their Haskell code.

ACKNOWLEDGMENTS

REFERENCES

[1] William E Byrd, Michael Ballantyne, Gregory Rosenblatt, and Matthew Might. 2017. A unified approach to solving seven programming problems (functional pearl). *Proceedings of the ACM on Programming Languages* 1, ICFP (2017).
[2] Daniel P. Friedman, William E. Byrd, Oleg Kiselyov, and Jason Hemann. 2018. *The Reasoned Schemer, Second Edition.*
[3] Oleg Kiselyov, Chung-chieh Shan, Daniel P Friedman, and Amr Sabry. 2005. Backtracking, interleaving, and terminating monad transformers:(functional pearl). *ACM SIGPLAN Notices* 40, 9 (2005), 192–203.
[4] Dmitri Rozplokhas and Dmitri Boulytchev. 2018. Improving Refutational Completeness of Relational Search via Divergence Test. In *Proceedings of the 20th International Symposium on Principles and Practice of Declarative Programming*. ACM, 18.
[5] Silvija Seres, J Michael Spivey, and C. A. R. Hoare. 1999. Algebra of Logic Programming.. In *ICLP*. 184–199.
[6] Edward Z. Yang. 2010. Adventures in Three Monads. *The Monad. Reader Issue 15* (2010), 11.

[c1] *DPF*: what
[c2] *LKC*: fixed